Problem:
Score : 400 points

Problem Statement

Find the largest integer that can be formed with exactly N matchsticks, under the following conditions:

Every digit in the integer must be one of the digits A_1, A_2, ..., A_M (1 \leq A_i \leq 9).

The number of matchsticks used to form digits 1, 2, 3, 4, 5, 6, 7, 8, 9 should be 2, 5, 5, 4, 5, 6, 3, 7, 6, respectively.

Constraints

All values in input are integers.

2 \leq N \leq 10^4

1 \leq M \leq 9

1 \leq A_i \leq 9

A_i are all different.

There exists an integer that can be formed by exactly N matchsticks under the conditions.

Input

Input is given from Standard Input in the following format:

N M
A_1 A_2 ... A_M

Output

Print the largest integer that can be formed with exactly N matchsticks under the conditions in the problem statement.

Sample Input 1

20 4
3 7 8 4

Sample Output 1

777773

The integer 777773 can be formed with 3 + 3 + 3 + 3 + 3 + 5 = 20 matchsticks, and this is the largest integer that can be formed by 20 matchsticks under the conditions.

Sample Input 2

101 9
9 8 7 6 5 4 3 2 1

Sample Output 2

71111111111111111111111111111111111111111111111111

The output may not fit into a 64-bit integer type.

Sample Input 3

15 3
5 4 6

Sample Output 3

654

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

program match_matching
  implicit none
  integer              :: n, m, b(9), i, j, d, r
  integer, allocatable :: a(:), dp(:)
  character(16384)     :: answer
  read(*,*) n, m
  allocate(a(m),dp(0:n))
  read(*,*) a(:)
  b = (/2,5,5,4,5,6,3,7,6/)
  call reverse_sort(m,a)
  dp = -1
  dp(0) = 0
  do i = 2, n
    d = -1
    do j = 1, m
      if (i-b(a(j)) .ge. 0) then
        if (dp(i-b(a(j))) .gt. d) d = dp(i-b(a(j)))
      end if
      if (d .lt. 0) then
        dp(i) = -1
      else
        dp(i) = d + 1
      end if
    end do
  end do
  r = n
  i = 1
  answer = ''
  do while (r .gt. 0)
    do j = 1, m
      if (r-b(a(j)) .ge. 0) then
        if (dp(r-b(a(j))) .eq. dp(r)-1) then
          answer(i:i) = achar(48+a(j))
          r = r - b(a(j))
          i = i + 1
          exit
        end if
      end if
    end do
  end do
  deallocate(a,dp)
  write(*,'(a)') answer(1:i)
  stop

contains

  subroutine reverse_sort(m,a)
    implicit none
    integer, intent(in)    :: m
    integer, intent(inout) :: a(m)
    integer                :: i, j
    do i = m-1, 1, -1
      do j = 1, i
        if (a(j) .lt. a(j+1)) call swap(a(j),a(j+1))
      end do
    end do
    return
  end subroutine reverse_sort

  subroutine swap(a,b)
    implicit none
    integer, intent(inout) :: a, b
    integer                :: c
    c = a
    a = b
    b = c
    return
  end subroutine swap

end program match_matching